- Existence-Unicité :

  Φ est une forme linéaire et $(P,Q) \mapsto \langle P,Q\rangle = \sum_{k=0}^{n} P(k)Q(k)$ est un produit scalaire sur $E = \mathbb{R}_n[X]$ donc il existe un $P \in E$ et un seul tel que $\Phi = \langle P,.\rangle$ (isomorphisme $a \in E \leftrightarrow (x \mapsto \langle a,x\rangle) \in E^*$).
- Existence-Unicité et expression de P_n :

  Pour tout $P \in E = \mathbb{R}_n[X]$, $\psi_P = (Q \mapsto \sum_{k=0}^{n} P(k)Q(k)$ est une forme linéaire sur E ainsi que Φ donc, pour toute base $\mathcal{B}$ de E :

  $$\psi_P = \Phi \iff \forall Q \in \mathcal{B},\ \psi_P(Q) = \Phi(Q),$$

  en particulier, en prenant pour $\mathcal{B}$ la base de Lagrange $\left(L_{n,k} = \prod_{0\leq i\leq n, i\neq k} \frac{X-i}{k-i}\right)_{(k=0..n)}$,

  $$\psi_P = \Phi \iff \forall k \in [\![0,n]\!],\ \psi_P(L_{n,k}) = \Phi(L_{n,k}) \iff \forall k \in [\![0,n]\!],\ P(k) = \Phi(L_{n,k}),$$

  et on sait que la décomposition de tout $P \in E$ dans la base de Lagrange $(L_{n,k})$ est $P = \sum_{k=0}^{n} P(k)L_{n,k}$ donc $\psi_P = \Phi$ a une solution et une seule, qui est $P = \sum_{k=0}^{n} \Phi(L_{n,k})L_{n,k}$.

O19-105

```
> restart:
```

Méthode 1 : on raisonne pour trouver l'expression de P_n au moyen des polynômes de Lagrange

```
> Lag:=(n,k)->product((t-i)/(k-i),i=0..(k-1))*product((t-i)/(k-i),
  i=(k+1)..n);
```

$$Lag := (n, k) \rightarrow \left(\prod_{i=0}^{k-1} \frac{t-i}{k-i}\right)\left(\prod_{i=k+1}^{n} \frac{t-i}{k-i}\right)$$

```
> Phi:=f->int(f,t=0..1);
```

$$\Phi := f \rightarrow \int_0^1 f \, dt$$

```
> P:=n->sum('Phi(Lag(n,k))*Lag(n,k)','k'=0..n);
```

$$P := n \rightarrow \sum_{'k'=0}^{n} '\Phi(\mathrm{Lag}(n,k))\,\mathrm{Lag}(n,k)'$$

```
> L:=[seq(sort(expand(P(n))),n=0..5)];
```

$$L := \left[1, \frac{1}{2}, -\frac{1}{2}t^2+\frac{3}{4}t+\frac{5}{12}, \frac{4}{9}t^3-\frac{49}{24}t^2+\frac{145}{72}t+\frac{3}{8}, -\frac{109}{432}t^4+\frac{1825}{864}t^3-\frac{1183}{216}t^2+\frac{3599}{864}t+\frac{251}{720}, \frac{53}{450}t^5-\frac{26413}{17280}t^4+\frac{60481}{8640}t^3-\frac{45517}{3456}t^2+\frac{178031}{21600}t+\frac{95}{288}\right]$$

Méthode 2 : on tente de résoudre le problème "Phi(Q)=sum(Q(k)P(k),k=0..n) pour Q=1,X,X^2,...,X^n" ie "1/(i+1)=sum(k^i.P(k),k=0..n) pour i=0,1,..,n".

```
> p:=proc(n)
  local s,sys,sol;
  s:=sum('a[k]*t^(k-1)','k'=1..(n+1));
  sys:={seq(sum('subs(t=j,s)*j^i','j'=0..n)-1/(i+1),i=0..n)};
  sol:=solve(sys,{seq(a[k],k=1..(n+1))});
  return subs(sol,s);
  end;
```

p := **proc**(n)
local s, sys, sol;
    s := sum('$a[k]$*t^($k-1$)', 'k' = 1 .. $n+1$);
    sys := { seq(sum('subs($t = j$, s)*j^i', 'j' = 0 .. n) − 1 / (i + 1), i = 0 .. n) };
    sol := solve(sys, { seq($a[k]$, k = 1 .. n + 1) });
    return subs(sol, s)
end proc

```
> l:=[seq(sort(p(n)),n=0..5)];ll:=[seq(p(n),n=2..5)]:
```

$$l := \left[1, \frac{1}{2}, -\frac{1}{2}t^2+\frac{3}{4}t+\frac{5}{12}, \frac{4}{9}t^3-\frac{49}{24}t^2+\frac{145}{72}t+\frac{3}{8}, -\frac{109}{432}t^4+\frac{1825}{864}t^3-\frac{1183}{216}t^2+\frac{3599}{864}t+\frac{251}{720}, \frac{53}{450}t^5-\frac{26413}{17280}t^4+\frac{60481}{8640}t^3-\frac{45517}{3456}t^2+\frac{178031}{21600}t+\frac{95}{288}\right]$$

```
> plot(ll,t=-1..5,legend=[seq(convert(i,string),i=2..5)]);
```


10
t
-1
0
1
2
3
4
5
-10
-20
-30
2
3
4

>